

```

/*
3. ČÁST
Výsledkem opakované bodové konstrukce
je množina bodů, které spojíme do lomené čáry.
Problém je v tom, že body nejdou za sebou, ale
napřeskáčku. Je to způsobeno tím, že v jednotlivé
bodové konstrukci vznikly dva body.
Potřebujeme proto skupiny uspořádat, tak aby šly
body za sebou. K tomu existuje příkaz usp, jehož
podrobnější popis naleznete v návodu.
Tuto skupinu bodu už lze spojit do jedné lomené
čáry.
*/

#vloz "zakladni.dk"

// průnik s rovinou ve výšce z
rez(z) {
    X = S + bod(0,0,z);
    alfa = rovnobezna(Pi, X);
    Y = p(alfa.2, (b:0).2); // průnik s bokem
    d = v(X.2, Y.2); // poloměr nové kružnice
    l = kruz(alfa, X, d); // řez kužele rovinou alfa

    r = p(ro, alfa); // průsečnice roviny řezu a alfa
    R = p(l, r); // výsledek bodové konstrukce

    vrat R;
}

hlavni () {
    zobrazeni("mongeovo");
    papir(29.7, 21, 1); //A4
    okraj(0.75,0.75,0.75,0.75,0.25,1,0);
    posunutipss(0,0);
    nadpis("", "", datum(), "", "ØEZ KUŽELE ROVINOU");

    // předdefinované styly
    normal = styl();

    sklopeny = styl();
    videt(sklopeny, vse, 2, ?, ?, 1, 3);
    barva(sklopeny, "seda");

    silny = styl();
    videt(silny, vse, 1, 0, 3, 2, 0);

    osa = styl();
    videt(osa, vse, 1, 0, 3, 1, 2);

    schovany = styl();
    videt(schovany, vse, 1, 0, 2, 1, 3);

    nevidet = styl();
    videt(nevidet, vse, 0);

    //kuzel
    S = bod(2,6,0); // střed podstavu
    r = 4; // poloměr podstavu
    v = 9; // výška kužele

```

```

k = kruz(Pi, S, r); // podstava
o = kolmice(Pi, S); // osa
V = S + bod(0,0,v); // vrchol

// boční čáry
K = bod(x, S, r);
b = usecka(K, V);

styl(o, osa);
styl(m(k, b), silny);

videt(b, 1, 0);
videt(K, 1, 0);
popis(K, 2, 0);

// rovina
ro = rovina(-9,9,8);

i = 0; // začínáme zdola
krok = 0.1; // přesnost

dokud (i < v-krok) { // vrchol vynecháme
    R = rez(i);
    zvetsi(i, krok);
}
zrus(i);
popis(R, vse, 0);

// uspořádání ze dvou částí, druhá je obrácené (-2)
// beznic(R) odstraní objekty typu nic ze skupiny R
R2 = usp(2, 1, -2, beznic(R));
zrus(R);
videt(R2, vse, 0);

// spojení bodů
rez = nuhelnik(R2);
}

```